
    BullMQ for Elixir

A powerful, fast, and robust job queue for Elixir backed by Redis. This is an Elixir port of the popular BullMQ library for Node.js, providing full compatibility with existing BullMQ queues.
Features
	⚡ High Performance - Built on Redis for speed and reliability
	🔄 Automatic Retries - Configurable retry strategies with exponential backoff
	⏰ Job Scheduling - Delay jobs or schedule them with cron expressions
	📊 Priority Queues - Process important jobs first
	🚦 Rate Limiting - Control processing rates
	👨‍👩‍👧‍👦 Parent-Child Jobs - Create complex workflows with dependencies
	📡 Real-time Events - Subscribe to job lifecycle events via Worker callbacks or QueueEvents
	🔒 Reliable - Stalled job detection and recovery
	📈 Observable - Built-in Telemetry integration
	🏗️ OTP Native - Built with GenServers and Supervisors
	🔄 Node.js Compatible - Jobs can be shared between Elixir and Node.js workers

Installation
Add bullmq to your list of dependencies in mix.exs:
def deps do
  [
    {:bullmq, "~> 1.0"}
  ]
end
Quick Start
1. Add Jobs to a Queue
# Add a job using stateless API (recommended for most use cases)
{:ok, job} = BullMQ.Queue.add("emails", "send-welcome", %{
  to: "user@example.com",
  template: "welcome"
}, connection: :my_redis)

# Add a delayed job
{:ok, job} = BullMQ.Queue.add("emails", "reminder", %{message: "Don't forget!"},
  connection: :my_redis,
  delay: 60_000  # 1 minute
)

# Add a prioritized job
{:ok, job} = BullMQ.Queue.add("emails", "urgent", %{},
  connection: :my_redis,
  priority: 1  # Lower = higher priority
)
2. Process Jobs with a Worker
defmodule MyApp.EmailWorker do
  def process(%BullMQ.Job{name: "send-welcome", data: data}) do
    MyApp.Mailer.send_welcome(data["to"], data["template"])
    {:ok, %{sent: true}}
  end

  def process(%BullMQ.Job{name: name}) do
    {:error, "Unknown job type: #{name}"}
  end
end

# Start a worker with event callbacks
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "emails",
  connection: :my_redis,
  processor: &MyApp.EmailWorker.process/1,
  concurrency: 5,
  on_completed: fn job, result ->
    IO.puts("Job #{job.id} completed with #{inspect(result)}")
  end,
  on_failed: fn job, reason ->
    IO.puts("Job #{job.id} failed: #{reason}")
  end
)
3. Add to Your Supervision Tree
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      # Start Redis connection
      {Redix, name: :my_redis, host: "localhost", port: 6379},

      # Start email worker
      {BullMQ.Worker,
        queue: "emails",
        connection: :my_redis,
        processor: &MyApp.EmailWorker.process/1,
        concurrency: 5
      }
    ]

    Supervisor.start_link(children, strategy: :one_for_one)
  end
end
Advanced Features
Job Options
BullMQ.Queue.add("tasks", "process-data", %{data: "..."},
  connection: :my_redis,
  priority: 1,              # Lower = higher priority
  delay: 60_000,            # Delay 60 seconds
  attempts: 5,              # Retry up to 5 times
  backoff: %{
    type: "exponential",
    delay: 1000
  },
  remove_on_complete: true, # Clean up after completion
  remove_on_fail: 100       # Keep last 100 failed jobs
)
Worker Event Callbacks
Workers support event callbacks similar to Node.js:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "tasks",
  connection: :my_redis,
  processor: &process/1,
  on_completed: fn job, result -> handle_completion(job, result) end,
  on_failed: fn job, reason -> handle_failure(job, reason) end,
  on_active: fn job -> handle_active(job) end,
  on_stalled: fn job_id -> handle_stalled(job_id) end
)
Queue Events (Real-time Subscriptions)
Subscribe to queue-level events using Redis Streams:
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "tasks",
  connection: :my_redis
)

BullMQ.QueueEvents.subscribe(events)

receive do
  {:bullmq_event, :completed, %{"jobId" => id}} ->
    IO.puts("Job #{id} completed!")
  {:bullmq_event, :failed, %{"jobId" => id, "failedReason" => reason}} ->
    IO.puts("Job #{id} failed: #{reason}")
end
Rate Limiting
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "api-calls",
  connection: :my_redis,
  processor: &process/1,
  limiter: %{max: 100, duration: 60_000}  # 100 per minute
)
Job Schedulers (Repeatable Jobs)
# Create a scheduler with cron pattern
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "maintenance", "cleanup",
  %{pattern: "0 * * * *"},  # Every hour
  "cleanup-job",
  %{type: "hourly"},
  prefix: "bull"
)

# Create an interval-based scheduler
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "heartbeats", "ping",
  %{every: 60_000},  # Every minute
  "heartbeat",
  %{},
  prefix: "bull"
)

# List schedulers
{:ok, schedulers} = BullMQ.JobScheduler.list(:my_redis, "maintenance", prefix: "bull")

# Remove a scheduler
{:ok, removed} = BullMQ.JobScheduler.remove(:my_redis, "maintenance", "cleanup", prefix: "bull")
Job Progress
def process(%BullMQ.Job{} = job) do
  Enum.each(1..100, fn i ->
    do_work(i)
    BullMQ.Worker.update_progress(job, i)
  end)

  {:ok, "done"}
end
Queue Getters
# Get job counts
{:ok, counts} = BullMQ.Queue.get_counts("emails", connection: :my_redis)
# => %{waiting: 10, active: 2, delayed: 5, completed: 100, failed: 3, ...}

# Get jobs in a specific state
{:ok, jobs} = BullMQ.Queue.get_jobs("emails", [:waiting, :delayed],
  connection: :my_redis, start: 0, end: 9)

# Get a specific job
{:ok, job} = BullMQ.Queue.get_job("emails", "job-id-123", connection: :my_redis)

# Get job state
{:ok, state} = BullMQ.Queue.get_job_state("emails", "job-id-123", connection: :my_redis)
# => :waiting | :active | :delayed | :completed | :failed
Queue Operations
# Pause the queue
:ok = BullMQ.Queue.pause("emails", connection: :my_redis)

# Resume the queue
:ok = BullMQ.Queue.resume("emails", connection: :my_redis)

# Check if paused
{:ok, is_paused} = BullMQ.Queue.paused?("emails", connection: :my_redis)

# Drain the queue (remove all waiting jobs)
:ok = BullMQ.Queue.drain("emails", connection: :my_redis)

# Remove a specific job
:ok = BullMQ.Queue.remove_job("emails", "job-id-123", connection: :my_redis)

# Retry a failed job
:ok = BullMQ.Queue.retry_job("emails", "job-id-123", connection: :my_redis)
Graceful Shutdown
Workers automatically wait for active jobs to complete when closing:
# Close worker and wait for active jobs to finish
:ok = BullMQ.Worker.close(worker)

# Force close without waiting
:ok = BullMQ.Worker.close(worker, force: true)
Documentation
Full documentation is available at HexDocs.
	Getting Started
	Workers
	Job Options
	Queue Events
	Rate Limiting
	Job Schedulers
	Telemetry

Requirements
	Elixir 1.15+
	Erlang/OTP 26+
	Redis 6.0+

Compatibility
This library is fully compatible with the Node.js BullMQ library. Jobs can be added from Node.js and processed by Elixir workers, and vice versa. They share the same Redis data structures and Lua scripts.
License
MIT License - see LICENSE for details.
Contributing
Contributions are welcome! Please see our Contributing Guide.
Commit Convention
This project uses Conventional Commits with automated releases via semantic-release. For Elixir-specific changes, add [elixir] tag to your commit message:
# Bug fix (patch release: 0.0.x)
git commit -m "fix(worker): handle job timeout correctly [elixir]"

# New feature (minor release: 0.x.0)
git commit -m "feat(queue): add bulk job operations [elixir]"

# Breaking change (major release: x.0.0)
git commit -m "feat(worker)!: change processor callback signature [elixir]"

	Commit Type	Version Bump	Example
	fix(...): ... [elixir]	Patch	fix(scripts): correct ARGV order [elixir]
	feat(...): ... [elixir]	Minor	feat(queue): add getJobCounts [elixir]
	feat(...)!: ... [elixir]	Major	feat(worker)!: new API [elixir]
	docs(...): ... [elixir]	None	docs(readme): update examples [elixir]
	chore(...): ... [elixir]	None	chore(deps): update redix [elixir]

Credits
This is an Elixir port of BullMQ by Taskforce.sh.


  

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
[Unreleased]
[1.0.0] - 2025-12-04
Added
	Initial release of BullMQ for Elixir
	Core queue functionality (BullMQ.Queue)	Add jobs with add/3 and add_bulk/3
	Pause and resume queues
	Get job by ID
	Drain and obliterate queues


	Worker implementation (BullMQ.Worker)	Configurable concurrency
	Automatic lock renewal
	Graceful shutdown
	Rate limiting support


	Job features (BullMQ.Job)	Priority queues
	Delayed jobs
	Automatic retries with backoff
	Progress tracking
	Custom job IDs


	Backoff strategies (BullMQ.Backoff)	Fixed backoff
	Exponential backoff
	Custom backoff functions
	Jitter support


	Rate limiting (BullMQ.RateLimiter)	Queue-level rate limits
	Group-based rate limits
	Manual rate limit triggering


	Job scheduling (BullMQ.JobScheduler)	Cron-based scheduling
	Interval-based scheduling
	Scheduler management (upsert, remove, list)


	Flow producer (BullMQ.FlowProducer)	Parent-child job dependencies
	Nested flows
	Bulk flow creation


	Stalled job detection (BullMQ.StalledChecker)	Automatic recovery
	Configurable stall limits


	Event streaming (BullMQ.QueueEvents)	Real-time job lifecycle events
	Event filtering


	Telemetry integration (BullMQ.Telemetry)	Job lifecycle events
	Worker events
	Rate limit events
	Span-based tracing


	Configuration validation (BullMQ.Config)	NimbleOptions-based schemas
	Queue, worker, and connection validation


	Redis key management (BullMQ.Keys)	Consistent key naming
	Configurable prefix


	Lua script execution (BullMQ.Scripts)	Atomic operations
	SHA caching
	Fallback to EVAL


	Redis connection pooling (BullMQ.RedisConnection)	NimblePool-based pooling
	Configurable pool size


	Comprehensive documentation	Getting started guide
	Job options reference
	Worker configuration
	Rate limiting guide
	Flow patterns
	Telemetry setup


	Test suite	Unit tests for all modules
	Integration tests (requires Redis)



Compatibility
	Compatible with Node.js BullMQ v5.x
	Requires Elixir 1.15+
	Requires Erlang/OTP 26+
	Requires Redis 6.0+



  

    Introduction

BullMQ is a robust, feature-rich message queue and job scheduling library for Elixir, built on top of Redis. It's a port of the popular BullMQ library from the Node.js ecosystem, providing full compatibility with existing BullMQ queues.
Features
	High Performance: Leverages Redis for fast, reliable message passing
	Job Scheduling: Schedule jobs to run at specific times or intervals
	Priority Queues: Process high-priority jobs first
	Retry Strategies: Automatic retries with configurable backoff
	Rate Limiting: Control job processing rates
	Parent-Child Jobs: Create complex job hierarchies with dependencies
	Real-time Events: Subscribe to job lifecycle events via Worker callbacks or QueueEvents
	Concurrency Control: Process multiple jobs simultaneously
	Stalled Job Recovery: Automatically recover jobs from crashed workers
	Telemetry Integration: Built-in observability with Telemetry
	OTP Design: Built using GenServers, Supervisors, and other OTP patterns
	Node.js Compatibility: Share queues between Elixir and Node.js workers

Quick Start
Add BullMQ to your dependencies:
def deps do
  [
    {:bullmq, "~> 1.0"}
  ]
end
Add jobs to a queue:
# Add a job (stateless API)
{:ok, job} = BullMQ.Queue.add("emails", "send-welcome", %{
  to: "user@example.com",
  template: "welcome"
}, connection: :my_redis)
Process jobs with a worker:
defmodule MyApp.EmailWorker do
  def process(%BullMQ.Job{name: "send-welcome", data: data}) do
    MyApp.Mailer.send_welcome(data["to"], data["template"])
    {:ok, %{sent: true}}
  end
end

# Start the worker with event callbacks
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "emails",
  connection: :my_redis,
  processor: &MyApp.EmailWorker.process/1,
  concurrency: 5,
  on_completed: fn job, result ->
    Logger.info("Job #{job.id} completed")
  end
)
Architecture
BullMQ uses Redis data structures to implement a reliable, distributed job queue:
	Lists for FIFO job queues (waiting, active)
	Sorted Sets for priority queues, delayed jobs, and rate limiting
	Hashes for job data storage
	Streams for real-time event delivery
	Lua Scripts for atomic operations

The Elixir port leverages OTP patterns:
	GenServer for stateful components (Worker, QueueEvents)
	Supervisor for fault tolerance
	Telemetry for observability
	True Parallelism using multiple BEAM processes for concurrent job processing

API Design
BullMQ for Elixir provides both stateless and stateful APIs:
Stateless API (Recommended)
Most queue operations work as simple function calls with a connection:
# Add a job
{:ok, job} = BullMQ.Queue.add("my_queue", "job_name", %{data: "value"},
  connection: :my_redis)

# Get job counts
{:ok, counts} = BullMQ.Queue.get_counts("my_queue", connection: :my_redis)

# Pause queue
:ok = BullMQ.Queue.pause("my_queue", connection: :my_redis)
Stateful API (GenServer)
Workers and QueueEvents run as supervised processes:
# Worker as GenServer
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "my_queue",
  connection: :my_redis,
  processor: &process/1
)

# QueueEvents as GenServer
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "my_queue",
  connection: :my_redis
)
Next Steps
	Read the Getting Started guide
	Learn about Workers
	Explore Job Options
	Understand Queue Events
	Set up Job Schedulers
	Configure Rate Limiting
	Add Telemetry



  

    Getting Started

This guide walks you through setting up BullMQ in your Elixir application.
Prerequisites
	Elixir 1.15 or later
	Erlang/OTP 26 or later
	Redis 6.0 or later

Installation
Add BullMQ to your mix.exs:
def deps do
  [
    {:bullmq, "~> 1.0"},
    {:redix, "~> 1.2"}  # Redis client
  ]
end
Then run:
mix deps.get

Setting Up Redis Connection
BullMQ uses Redix for Redis connections. Add a Redix connection to your supervision tree:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      # Redis connection
      {Redix, name: :my_redis, host: "localhost", port: 6379}
    ]

    Supervisor.start_link(children, strategy: :one_for_one)
  end
end
For production with authentication:
{Redix,
  name: :my_redis,
  host: System.get_env("REDIS_HOST", "localhost"),
  port: String.to_integer(System.get_env("REDIS_PORT", "6379")),
  password: System.get_env("REDIS_PASSWORD")
}
Adding Your First Job
Jobs are added using the BullMQ.Queue.add/4 function:
# Add a simple job
{:ok, job} = BullMQ.Queue.add("notifications", "push-notification", %{
  user_id: 123,
  message: "You have a new message!"
}, connection: :my_redis)

IO.puts("Created job: #{job.id}")
# => Created job: 5f8a9b2c3d4e5f6a7b8c9d0e
The function takes:
	Queue name (string)
	Job name/type (string)
	Job data (map)
	Options (keyword list with :connection required)

Creating Your First Worker
A worker processes jobs from a queue:
defmodule MyApp.NotificationWorker do
  alias BullMQ.Job

  def process(%Job{name: "push-notification", data: data}) do
    user = MyApp.Users.get(data["user_id"])
    MyApp.PushService.send(user.device_token, data["message"])

    {:ok, %{sent: true}}
  end

  def process(%Job{name: name}) do
    {:error, "Unknown job type: #{name}"}
  end
end
Start the worker:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "notifications",
  connection: :my_redis,
  processor: &MyApp.NotificationWorker.process/1,
  concurrency: 10
)
Complete Application Setup
Here's a complete supervision tree setup:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      # Redis connection
      {Redix, name: :my_redis, host: "localhost", port: 6379},

      # Notification worker
      {BullMQ.Worker,
        queue: "notifications",
        connection: :my_redis,
        processor: &MyApp.NotificationWorker.process/1,
        concurrency: 10,
        on_completed: fn job, result ->
          Logger.info("Job #{job.id} completed: #{inspect(result)}")
        end,
        on_failed: fn job, reason ->
          Logger.error("Job #{job.id} failed: #{reason}")
        end
      },

      # Email worker
      {BullMQ.Worker,
        queue: "emails",
        connection: :my_redis,
        processor: &MyApp.EmailWorker.process/1,
        concurrency: 5
      }
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
Job Lifecycle
Jobs go through several states:
	waiting - Job is waiting to be picked up
	active - Job is being processed
	completed - Job finished successfully
	failed - Job failed after all retries
	delayed - Job is scheduled for future execution

# Check job state
{:ok, state} = BullMQ.Queue.get_job_state("notifications", job.id,
  connection: :my_redis)

IO.puts("Job state: #{state}")
# => Job state: completed
Processor Return Values
Your processor function should return one of:
def process(job) do
  case do_work(job.data) do
    # Success with result - job moves to completed
    {:ok, result} ->
      {:ok, result}

    # Simple success
    :ok ->
      :ok

    # Failure - triggers retry if attempts remaining, otherwise moves to failed
    {:error, reason} ->
      {:error, reason}
  end
end

# Raising an exception also triggers retry (if attempts remaining)
def process(job) do
  if something_wrong?(job) do
    raise "Something went wrong"
  end
  :ok
end
Both {:error, reason} and raising exceptions trigger the same retry behavior.
See Workers for more details on return values.
Adding Jobs with Options
Customize job behavior with options:
# Delayed job
BullMQ.Queue.add("emails", "send-reminder", %{user_id: 123},
  connection: :my_redis,
  delay: 60_000  # 1 minute delay
)

# Priority job
BullMQ.Queue.add("emails", "urgent-alert", %{},
  connection: :my_redis,
  priority: 1  # Lower = higher priority
)

# Job with retries
BullMQ.Queue.add("api-sync", "sync-data", %{},
  connection: :my_redis,
  attempts: 5,
  backoff: %{type: "exponential", delay: 1000}
)
Queue Prefix
By default, BullMQ uses "bull" as a prefix for all Redis keys. You can customize this:
# When adding jobs
BullMQ.Queue.add("emails", "send", %{},
  connection: :my_redis,
  prefix: "myapp"
)

# When starting workers
BullMQ.Worker.start_link(
  queue: "emails",
  connection: :my_redis,
  prefix: "myapp",
  processor: &process/1
)
Important: All components accessing the same queue must use the same prefix.
Next Steps
	Learn about Workers for advanced processing options
	Explore Job Options for customizing job behavior
	Set up Queue Events for real-time monitoring
	Configure Rate Limiting to control throughput
	Create recurring jobs with Job Schedulers
	Use Job Flows for complex workflows
	Monitor with Telemetry



  

    Job Options

BullMQ provides extensive options for customizing job behavior.
Adding Jobs
Jobs are added using the BullMQ.Queue.add/4 function:
{:ok, job} = BullMQ.Queue.add(queue_name, job_name, data, opts)
Where:
	queue_name - The queue name (string)
	job_name - Job type/name for pattern matching (string)
	data - Job payload (map)
	opts - Options including :connection and job-specific options

Priority
Jobs with lower priority values are processed first:
# High priority job (processed first)
BullMQ.Queue.add("tasks", "urgent-task", %{},
  connection: :my_redis,
  priority: 1
)

# Normal priority (default)
BullMQ.Queue.add("tasks", "normal-task", %{},
  connection: :my_redis
)

# Low priority job (processed last)
BullMQ.Queue.add("tasks", "batch-task", %{},
  connection: :my_redis,
  priority: 100
)
Priority uses a Redis sorted set, so jobs are always processed in priority order.
Delay
Schedule jobs to run after a delay:
# Run in 5 minutes
BullMQ.Queue.add("reminders", "send-reminder", %{message: "Don't forget!"},
  connection: :my_redis,
  delay: 5 * 60 * 1000  # 5 minutes in milliseconds
)

# Run at a specific time
future_time = DateTime.utc_now() |> DateTime.add(3600, :second)
delay = DateTime.diff(future_time, DateTime.utc_now(), :millisecond)

BullMQ.Queue.add("reports", "scheduled-report", %{},
  connection: :my_redis,
  delay: delay
)
Retries and Backoff
Configure automatic retry behavior:
# 3 retries with exponential backoff
BullMQ.Queue.add("api-calls", "call-api", %{url: "..."},
  connection: :my_redis,
  attempts: 3,
  backoff: %{type: "exponential", delay: 1000}
)
# Delays: 1s, 2s, 4s

# Fixed backoff
BullMQ.Queue.add("api-calls", "call-api", %{url: "..."},
  connection: :my_redis,
  attempts: 5,
  backoff: %{type: "fixed", delay: 5000}
)
# Delays: 5s, 5s, 5s, 5s
Backoff Types
	exponential - Delay doubles each attempt: delay * 2^attempt
	fixed - Same delay each time

Custom Job IDs
By default, jobs get a unique ID. You can specify a custom ID:
# Using custom job ID
BullMQ.Queue.add("users", "process-user", %{user_id: 123},
  connection: :my_redis,
  job_id: "user-123-process"
)

# Adding the same job ID again will return the existing job
Deduplication
Prevent duplicate jobs from being added to the queue. See the Deduplication Guide for full details.
# Simple mode: deduplicate until job completes
BullMQ.Queue.add("tasks", "process", %{},
  connection: :my_redis,
  deduplication: %{id: "unique-task-id"}
)

# Throttle mode: deduplicate for 5 seconds
BullMQ.Queue.add("tasks", "process", %{},
  connection: :my_redis,
  deduplication: %{id: "unique-task-id", ttl: 5_000}
)

# Debounce mode: replace and extend TTL
BullMQ.Queue.add("tasks", "process", %{data: "latest"},
  connection: :my_redis,
  delay: 5_000,
  deduplication: %{id: "unique-task-id", ttl: 5_000, extend: true, replace: true}
)
LIFO Processing
By default, jobs are processed FIFO (first in, first out). Use LIFO for stack-like behavior:
# This job will be processed before older jobs
BullMQ.Queue.add("urgent", "urgent-task", %{},
  connection: :my_redis,
  lifo: true
)
Job Cleanup
Control when completed/failed jobs are removed:
# Remove immediately when completed
BullMQ.Queue.add("temporary", "temp-job", %{},
  connection: :my_redis,
  remove_on_complete: true
)

# Keep last 100 completed jobs
BullMQ.Queue.add("with-history", "job", %{},
  connection: :my_redis,
  remove_on_complete: %{count: 100}
)

# Remove completed jobs older than 1 hour (in ms)
BullMQ.Queue.add("time-limited", "job", %{},
  connection: :my_redis,
  remove_on_complete: %{age: 3_600_000}
)

# Remove failed jobs after keeping 50
BullMQ.Queue.add("cleanup-failures", "job", %{},
  connection: :my_redis,
  remove_on_fail: %{count: 50}
)

# Keep completed jobs but remove failed ones
BullMQ.Queue.add("success-matters", "job", %{},
  connection: :my_redis,
  remove_on_complete: false,
  remove_on_fail: true
)
Bulk Operations
Add multiple jobs atomically using add_bulk/3. This function uses Redis MULTI/EXEC transactions to ensure all jobs are added atomically (all or nothing), achieving up to 10x higher throughput than individual add/4 calls.
Basic Usage
jobs = [
  {"email", %{to: "user1@example.com"}, [priority: 1]},
  {"email", %{to: "user2@example.com"}, []},
  {"email", %{to: "user3@example.com"}, [delay: 60_000]}
]

# All jobs are added atomically - either all succeed or none do
{:ok, added_jobs} = BullMQ.Queue.add_bulk("emails", jobs, connection: :my_redis)
High-Performance Bulk Addition
For adding large numbers of jobs (10,000+), use a connection pool for parallel processing:
# Create a pool of 8 connections
pool = for i <- 1..8 do
  name = :"redis_pool_#{i}"
  {:ok, _} = BullMQ.RedisConnection.start_link(name: name, host: "localhost")
  name
end

# Add 100,000 jobs at ~60,000 jobs/sec
# Each chunk is added atomically
jobs = for i <- 1..100_000, do: {"job", %{index: i}, []}

{:ok, added} = BullMQ.Queue.add_bulk("my-queue", jobs,
  connection: :redis,
  connection_pool: pool
)
Bulk Options
	Option	Default	Description
	pipeline	true	Use transactional pipelining (4x faster, atomic)
	chunk_size	100	Jobs per transaction batch
	connection_pool	nil	List of connections for parallel processing
	concurrency	8	Max parallel tasks

See Benchmarks for detailed performance data.
All Options Reference
	Option	Type	Default	Description
	connection	atom/pid	required	Redis connection
	prefix	string	"bull"	Queue prefix
	priority	integer	0	Lower = higher priority
	delay	integer	0	Delay in milliseconds
	attempts	integer	1	Total attempts including first
	backoff	map	nil	Retry strategy config
	lifo	boolean	false	Add to front of queue
	job_id	string	auto	Custom job identifier
	deduplication	map	nil	Deduplication config (see guide)
	remove_on_complete	bool/map	false	Cleanup config
	remove_on_fail	bool/map	false	Cleanup config
	keep_logs	integer	nil	Maximum log entries to keep
	timestamp	integer	now	Job creation timestamp
	telemetry_metadata	string	nil	Serialized trace context (auto-set by telemetry)
	omit_context	boolean	false	Skip trace context propagation

Next Steps
	Learn about Workers for processing jobs
	Set up Rate Limiting to control throughput
	Create recurring jobs with Job Schedulers
	Use Deduplication to prevent duplicate jobs
	Configure Telemetry for observability and distributed tracing



  

    Workers

Workers are the processes that fetch and execute jobs from queues.
Basic Worker
The simplest worker configuration:
defmodule MyApp.EmailWorker do
  def process(%BullMQ.Job{data: data}) do
    send_email(data["to"], data["subject"], data["body"])
    {:ok, %{sent: true}}
  end
end

{:ok, worker} = BullMQ.Worker.start_link(
  queue: "emails",
  connection: :my_redis,
  processor: &MyApp.EmailWorker.process/1
)
Worker Options
{:ok, worker} = BullMQ.Worker.start_link(
  # Required options
  queue: "my_queue",                    # Queue name
  connection: :my_redis,                # Redis connection (Redix name or pid)
  processor: &MyApp.Worker.process/1,   # Processor function

  # Optional options
  name: :my_worker,                     # Process registration name
  concurrency: 10,                      # Max concurrent jobs (default: 1)
  lock_duration: 30_000,                # Lock TTL in ms (default: 30000)
  stalled_interval: 30_000,             # Stalled check interval (default: 30000)
  max_stalled_count: 1,                 # Max stalls before failure (default: 1)
  prefix: "bull",                       # Queue prefix (default: "bull")
  autorun: true,                        # Start processing immediately (default: true)
  limiter: %{max: 100, duration: 60_000}, # Rate limiting config
  telemetry: BullMQ.Telemetry.OpenTelemetry, # OpenTelemetry integration (optional)

  # Event callbacks
  on_completed: fn job, result -> ... end,
  on_failed: fn job, reason -> ... end,
  on_active: fn job -> ... end,
  on_stalled: fn job_id -> ... end,
  on_error: fn error -> ... end,
  on_progress: fn job, progress -> ... end
)
Concurrency
Process multiple jobs simultaneously:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "image-processing",
  connection: :my_redis,
  processor: &MyApp.ImageProcessor.process/1,
  concurrency: 20  # Process 20 images at once
)
Each concurrent job runs in its own process, providing isolation and fault tolerance. Unlike Node.js which uses a single thread with async operations, Elixir workers use true parallelism with multiple BEAM processes.
Event Callbacks
Workers support event callbacks similar to Node.js worker.on('completed', ...):
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "tasks",
  connection: :my_redis,
  processor: &process/1,

  on_completed: fn job, result ->
    Logger.info("Job #{job.id} completed with result: #{inspect(result)}")
  end,

  on_failed: fn job, reason ->
    Logger.error("Job #{job.id} failed: #{reason}")
    MyApp.Alerts.notify("Job failed: #{reason}")
  end,

  on_active: fn job ->
    Logger.debug("Job #{job.id} started processing")
  end,

  on_stalled: fn job_id ->
    Logger.warning("Job #{job_id} stalled")
  end,

  on_error: fn error ->
    Logger.error("Worker error: #{inspect(error)}")
  end,

  on_progress: fn job, progress ->
    Logger.debug("Job #{job.id} progress: #{inspect(progress)}")
  end
)
Using Callbacks for Testing
Callbacks are particularly useful in tests for waiting on job completion without polling:
test "processes job successfully" do
  test_pid = self()

  {:ok, worker} = BullMQ.Worker.start_link(
    queue: "test-queue",
    connection: :my_redis,
    processor: fn job -> {:ok, job.data["value"] * 2} end,
    on_completed: fn job, result ->
      send(test_pid, {:completed, job.id, result})
    end
  )

  {:ok, job} = BullMQ.Queue.add("test-queue", "test", %{value: 21},
    connection: :my_redis)

  assert_receive {:completed, job_id, 42}, 5_000
  assert job_id == job.id
end
You can also test progress updates using the on_progress callback:
test "reports progress correctly" do
  test_pid = self()

  {:ok, worker} = BullMQ.Worker.start_link(
    queue: "test-queue",
    connection: :my_redis,
    processor: fn job ->
      BullMQ.Worker.update_progress(job, 50)
      BullMQ.Worker.update_progress(job, 100)
      :ok
    end,
    on_progress: fn job, progress ->
      send(test_pid, {:progress, job.id, progress})
    end,
    on_completed: fn _job, _result ->
      send(test_pid, :completed)
    end
  )

  {:ok, job} = BullMQ.Queue.add("test-queue", "test", %{},
    connection: :my_redis)

  job_id = job.id
  assert_receive {:progress, ^job_id, 50}, 5_000
  assert_receive {:progress, ^job_id, 100}, 5_000
  assert_receive :completed, 5_000
end
Processor Return Values
Processors can return different values to control job outcome:
	Return Value	Description
	{:ok, result}	Job completed successfully. Result is stored and on_completed callback fires.
	:ok	Job completed successfully (no result stored).
	{:error, reason}	Job failed. Triggers retry if attempts remain, otherwise moves to failed.
	{:delay, ms}	Move job to delayed state for ms milliseconds. Does not increment attempt count.
	{:rate_limit, ms}	Move job to delayed state due to rate limiting. Similar to :delay.
	:waiting	Move job back to waiting queue immediately.
	:waiting_children	Move job to waiting-children state (waits for child jobs to complete).

def process(job) do
  case do_work(job.data) do
    # Success - job moves to completed
    {:ok, result} ->
      {:ok, result}

    # Simple success (no return value stored)
    :ok ->
      :ok

    # Error that triggers retry (if attempts remaining)
    {:error, reason} ->
      {:error, reason}

    # Delay job for 5 seconds (without incrementing attempts)
    {:needs_delay, ms} ->
      {:delay, ms}

    # Manual rate limiting - delay due to external rate limit
    :rate_limited ->
      {:rate_limit, 60_000}

    # Move back to waiting queue (immediate retry by any worker)
    :should_wait ->
      :waiting

    # Wait for child jobs to complete
    :has_children ->
      :waiting_children
  end
end

# Raising an exception also triggers retry
def process(job) do
  if something_wrong?(job) do
    raise "Something went wrong"
  end
  :ok
end
When to Use Each Return Value
{:delay, ms} - Delay for Later Processing
Use when the job needs to wait before being processed again:
def process(job) do
  case check_resource_status(job.data["resource_id"]) do
    :ready ->
      process_resource(job.data)
      {:ok, :processed}

    :not_ready ->
      # Resource not ready yet, check again in 30 seconds
      {:delay, 30_000}

    :pending_approval ->
      # Wait for human approval, check every 5 minutes
      {:delay, 300_000}
  end
end
Key behaviors:
	Job moves to delayed queue for the specified duration
	Does NOT increment the attempt count
	Does NOT trigger on_completed callback
	Job will be picked up by any available worker after the delay

{:rate_limit, ms} - Manual Rate Limiting
Use when you detect rate limiting from an external service:
def process(job) do
  case MyApp.ExternalAPI.call(job.data) do
    {:ok, response} ->
      {:ok, response}

    {:error, :rate_limited, retry_after} ->
      # API told us to wait, respect it
      {:rate_limit, retry_after * 1000}

    {:error, 429, headers} ->
      # HTTP 429 - extract Retry-After header
      retry_ms = parse_retry_after(headers) || 60_000
      {:rate_limit, retry_ms}
  end
end
Key behaviors:
	Identical to {:delay, ms} in execution
	Semantically indicates rate limiting (useful for logging/monitoring)
	Does NOT increment the attempt count
	Does NOT trigger on_completed callback

:waiting - Return to Waiting Queue
Use when the job should be retried immediately by any worker:
def process(job) do
  case acquire_distributed_lock(job.data["resource"]) do
    {:ok, lock} ->
      result = do_work_with_lock(job.data, lock)
      release_lock(lock)
      {:ok, result}

    :locked_by_another ->
      # Another worker has the lock, let a different worker try
      # (maybe on a different node that has the lock)
      :waiting
  end
end

# Another use case: load balancing across workers
def process(job) do
  if worker_overloaded?() do
    # Let another worker handle this
    :waiting
  else
    do_work(job)
  end
end
Key behaviors:
	Job returns to waiting queue immediately (no delay)
	Will be picked up by the next available worker (possibly different node)
	Does NOT increment the attempt count
	Does NOT trigger on_completed callback
	Useful for distributed coordination scenarios

:waiting_children - Wait for Child Jobs
Use with parent-child job flows when the parent needs to wait for children:
def process(%{name: "process-batch"} = job) do
  # Create child jobs for each item in the batch
  Enum.each(job.data["items"], fn item ->
    BullMQ.FlowProducer.add_child(job, "process-item", item)
  end)

  # Wait for all children to complete before this job continues
  :waiting_children
end

def process(%{name: "process-item"} = job) do
  # Process individual item
  result = process_item(job.data)
  {:ok, result}
end
Key behaviors:
	Job moves to waiting-children state
	Automatically resumed when all child jobs complete
	Parent can access child results via BullMQ.Job.get_children_values/1
	Does NOT trigger on_completed callback (until children complete and job finishes)
	See Flows & Parent-Child Jobs for details

Comparison Summary
	Return Value	Queue State	Delay	Increment Attempts	on_completed
	{:ok, result}	completed	-	-	✅ Yes
	:ok	completed	-	-	✅ Yes
	{:error, reason}	delayed/failed	backoff	✅ Yes	❌ No (until final failure)
	{:error, reason}	delayed/failed	backoff	✅ Yes	❌ No (until final failure)
	{:delay, ms}	delayed	specified	❌ No	❌ No
	{:rate_limit, ms}	delayed	specified	❌ No	❌ No
	:waiting	waiting	none	❌ No	❌ No
	:waiting_children	waiting-children	none	❌ No	❌ No

Failures: {:error, reason} vs Exceptions
Both {:error, reason} return values and exceptions trigger the same retry behavior:
# These two are equivalent in terms of retry behavior:

# Option 1: Return error tuple (idiomatic Elixir)
def process(job) do
  case external_api_call(job.data) do
    {:ok, result} -> {:ok, result}
    {:error, reason} -> {:error, reason}  # Triggers retry
  end
end

# Option 2: Raise exception (Node.js style)
def process(job) do
  result = external_api_call!(job.data)  # Raises on error
  {:ok, result}
end
When to use each:
	Approach	Best For
	{:error, reason}	Expected failures from API calls, validation errors, pattern matching on results
	raise	Unexpected errors, assertion failures, "this should never happen" cases
	throw	Non-local returns (rare in typical code)
	exit	Process termination signals

All of these:
	Trigger retry if attempts remain
	Move job to failed after max retries
	Store the error reason/message with the job
	Call on_failed callback (only on final failure)

Exception Handling
If your processor raises an exception, exits, or throws a value, BullMQ catches it
automatically and treats it as a job failure. The worker process does NOT crash -
all errors are safely contained within the job processing context.
def process(job) do
  # If this raises, the job will fail (and retry if attempts remain)
  result = dangerous_operation!(job.data)
  {:ok, result}
end

def process(job) do
  # Explicit validation with raise
  if invalid?(job.data) do
    raise ArgumentError, "Invalid job data: #{inspect(job.data)}"
  end

  do_work(job.data)
end
All error types are caught:
# These all result in job failure:

# 1. raise - Elixir exceptions
def process(_job), do: raise "Something went wrong"

# 2. exit - Process exit signals
def process(_job), do: exit(:abnormal_termination)

# 3. throw - Thrown values (non-local returns)
def process(_job), do: throw(:abort_processing)
Exception behavior:
	Scenario	Result	Failure Reason
	raise "error"	Job fails, retries if attempts remaining	Exception message
	exit(:reason)	Job fails, retries if attempts remaining	Inspected reason
	throw(:value)	Job fails, retries if attempts remaining	Inspected value
	Timeout (lock expires)	Job becomes stalled, handled by stalled checker	N/A

What gets captured:
	Exception message via Exception.message(e) for raised exceptions
	Inspected value for exit and throw
	Full stacktrace (stored with the job for debugging)
	Exit reasons from linked processes

Example with retries:
# Add job with 3 attempts
{:ok, job} = BullMQ.Queue.add("emails", "send", %{to: "user@example.com"},
  connection: :redis,
  attempts: 3,
  backoff: %{type: :exponential, delay: 1000}
)

# In processor - exceptions trigger retry with backoff
def process(job) do
  case send_email(job.data) do
    :ok ->
      {:ok, :sent}

    {:error, :temporary_failure} ->
      # Explicit error - will retry
      {:error, "Temporary failure, will retry"}

    {:error, :permanent_failure} ->
      # You could also raise for permanent failures
      raise "Permanent failure: email address invalid"
  end
end
Viewing failure information:
When a job fails, the error message and stacktrace are stored:
{:ok, job} = BullMQ.Queue.get_job("my-queue", job_id, connection: :redis)

# Access failure info
job.failed_reason      # "ArgumentError: Invalid job data: %{...}"
job.stacktrace         # ["    (my_app 1.0.0) lib/my_app/worker.ex:15: ..."]
job.attempts_made      # Number of attempts so far
Best practices:
	Let it crash for unexpected errors - BullMQ handles retries automatically
	Use {:error, reason} for expected failures - More explicit control
	Configure appropriate retry attempts - Default is 0 (no retries)
	Use backoff strategies - Exponential backoff for transient failures
	Monitor failed jobs - Use on_failed callback or QueueEvents

{:ok, worker} = BullMQ.Worker.start_link(
  queue: "critical-jobs",
  connection: :redis,
  processor: &process/1,

  # Get notified when jobs exhaust all retries
  on_failed: fn job, reason ->
    Logger.error("Job #{job.id} failed permanently: #{reason}")
    MyApp.Alerts.notify("Critical job failed", job: job, reason: reason)
  end
)
Lock Duration
Workers hold a lock on jobs to prevent duplicate processing. If the lock expires before the job completes, another worker might pick it up:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "long-running",
  connection: :my_redis,
  processor: &MyApp.LongRunning.process/1,
  lock_duration: 300_000  # 5 minutes (default: 30 seconds)
)
BullMQ automatically renews locks at half the lock duration interval.
Automatic Cancellation on Lock Loss
If a lock renewal fails (e.g., due to network issues or Redis problems), the worker automatically cancels the affected job. This prevents duplicate processing if another worker picks up the job.
When a lock is lost, the processor receives a cancellation message with reason {:lock_lost, job_id}:
processor: fn job, cancel_token ->
  receive do
    {:cancel, ^cancel_token, {:lock_lost, _job_id}} ->
      # Lock was lost - stop processing to avoid duplicates
      Logger.warning("Lock lost for job #{job.id}, stopping")
      {:error, :lock_lost}
    {:cancel, ^cancel_token, reason} ->
      {:error, {:cancelled, reason}}
  after
    0 ->
      do_work(job)
  end
end
You can also use the on_lock_renewal_failed callback to be notified:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "my-queue",
  connection: :my_redis,
  processor: &MyApp.process/2,
  on_lock_renewal_failed: fn job_ids ->
    Logger.error("Lock renewal failed for jobs: #{inspect(job_ids)}")
    # Alert monitoring, etc.
  end
)
Stalled Job Recovery
Jobs can "stall" when a worker crashes or loses connection before completing a job.
BullMQ automatically detects and recovers stalled jobs.
Default Configuration
The stalled job detection has sensible defaults that should normally not be changed:
	Option	Default	Description
	lock_duration	30,000ms	Time before a job lock expires
	stalled_interval	30,000ms	How often to check for stalled jobs
	max_stalled_count	1	Times a job can stall before failing

Why max_stalled_count defaults to 1
Stalled jobs are considered a rare occurrence. If a job stalls more than once, it
typically indicates a more serious issue:
	Repeated worker crashes on specific job data
	Resource exhaustion (memory, CPU)
	External service failures
	Bugs in job processing logic

Instead of increasing max_stalled_count, investigate and fix the underlying issue.
Monitoring Stalled Jobs
Use the on_stalled callback to monitor when jobs stall:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "important",
  connection: :my_redis,
  processor: &MyApp.Important.process/1,
  on_stalled: fn job_id ->
    Logger.warning("Job #{job_id} stalled - investigating...")
    # Alert your monitoring system
  end
)
When to Adjust Settings
Only change these settings if you have a specific need:
# Only if jobs legitimately take > 30s between progress updates
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "long-running",
  connection: :my_redis,
  processor: &MyApp.LongJob.process/1,
  lock_duration: 120_000  # 2 minutes for very long jobs
)
Rate Limiting
Control how many jobs are processed per time window:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "api-calls",
  connection: :my_redis,
  processor: &MyApp.ApiCaller.process/1,
  limiter: %{
    max: 100,           # Max 100 jobs
    duration: 60_000    # Per minute
  }
)
See the Rate Limiting guide for more details.
Pattern Matching on Job Names
Process different job types with pattern matching:
defmodule MyApp.NotificationWorker do
  alias BullMQ.Job

  def process(%Job{name: "email", data: data}) do
    send_email(data)
  end

  def process(%Job{name: "sms", data: data}) do
    send_sms(data)
  end

  def process(%Job{name: "push", data: data}) do
    send_push_notification(data)
  end

  def process(%Job{name: name}) do
    {:error, "Unknown notification type: #{name}"}
  end
end
Job Progress
Report progress for long-running jobs:
def process(%BullMQ.Job{} = job) do
  items = fetch_items(job.data)
  total = length(items)

  items
  |> Enum.with_index()
  |> Enum.each(fn {item, index} ->
    process_item(item)

    # Report progress (any value - typically 0-100)
    progress = round((index + 1) / total * 100)
    BullMQ.Worker.update_progress(job, progress)
  end)

  {:ok, %{processed: total}}
end
Progress updates emit a progress event in Redis Streams and trigger the on_progress callback.
Job Logging
Add log entries to a job. Logs are stored in Redis and can be retrieved later for debugging or tracking progress.
You can use either Job.log/2 or Worker.log/2:
def process(%BullMQ.Job{} = job) do
  # Using Job.log (returns {:ok, log_count})
  {:ok, 1} = BullMQ.Job.log(job, "Starting processing")

  result = do_work(job.data)

  {:ok, 2} = BullMQ.Job.log(job, "Completed with result: #{inspect(result)}")
  {:ok, result}
end
Or using Worker.log/2:
def process(%BullMQ.Job{} = job) do
  # Using Worker.log (returns :ok)
  :ok = BullMQ.Worker.log(job, "Starting processing")

  result = do_work(job.data)

  :ok = BullMQ.Worker.log(job, "Completed with result: #{inspect(result)}")
  {:ok, result}
end
Limiting Log Entries
Use the keep_logs option to limit the number of log entries stored:
# Only keep the last 10 log entries
BullMQ.Job.log(job, "Processing step 1", keep_logs: 10)
BullMQ.Job.log(job, "Processing step 2", keep_logs: 10)
You can also set this globally when adding a job:
{:ok, job} = BullMQ.Queue.add("my-queue", "job-name", %{data: "value"},
  keep_logs: 100)
Graceful Shutdown
Workers automatically complete in-progress jobs before shutting down:
# Close the worker and wait for active jobs to complete
:ok = BullMQ.Worker.close(worker)

# Force close without waiting for jobs
:ok = BullMQ.Worker.close(worker, force: true)
For supervised workers, configure the shutdown timeout:
children = [
  %{
    id: MyWorker,
    start: {BullMQ.Worker, :start_link, [[
      queue: "jobs",
      connection: :my_redis,
      processor: &MyApp.process/1
    ]]},
    shutdown: 60_000  # Wait up to 60 seconds for jobs to complete
  }
]
Pause and Resume
Pause and resume job processing:
# Pause the worker (finishes current jobs, stops picking new ones)
:ok = BullMQ.Worker.pause(worker)

# Resume processing
:ok = BullMQ.Worker.resume(worker)

# Check if paused
BullMQ.Worker.paused?(worker)
# => true
Supervision
Add workers to your supervision tree for automatic restarts:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      # Redis connection
      {Redix, name: :my_redis, host: "localhost", port: 6379},

      # Email worker
      {BullMQ.Worker,
        name: :email_worker,
        queue: "emails",
        connection: :my_redis,
        processor: &MyApp.EmailWorker.process/1,
        concurrency: 5
      },

      # Heavy processing worker
      {BullMQ.Worker,
        name: :heavy_worker,
        queue: "heavy",
        connection: :my_redis,
        processor: &MyApp.HeavyWorker.process/1,
        concurrency: 2,
        lock_duration: 300_000
      }
    ]

    Supervisor.start_link(children, strategy: :one_for_one)
  end
end
Worker Telemetry
Workers emit telemetry events for observability and support OpenTelemetry for distributed tracing.
OpenTelemetry
Enable distributed tracing by passing a telemetry module:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "my-queue",
  connection: :my_redis,
  processor: &MyApp.Worker.process/1,
  telemetry: BullMQ.Telemetry.OpenTelemetry
)
When enabled, the worker automatically:
	Restores trace context from the job's tm (telemetry_metadata) option
	Creates spans linked to the producer's trace
	Records errors and exceptions on spans

See the Telemetry guide for full OpenTelemetry setup.
Next Steps
	Learn about Rate Limiting
	Create Job Flows
	Set up Telemetry



  

    Manual Job Processing

BullMQ Elixir supports manual job processing, where you have full control over fetching jobs and managing their lifecycle instead of using automatic worker processing.
Overview
Manual processing is useful when you need:
	Fine-grained control over job execution
	Custom job routing or filtering
	Integration with external systems that control processing flow
	Rate limiting at the application level
	Processing jobs in batches

Basic Usage
Creating a Worker for Manual Processing
Create a worker without automatic processing by setting autorun: false:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "my-queue",
  connection: :redis,
  processor: nil,  # No processor needed for manual processing
  autorun: false,  # Don't start automatic job fetching
  lock_duration: 30_000  # Lock duration for fetched jobs
)

# Start the stalled job checker (recommended)
:ok = BullMQ.Worker.start_stalled_check_timer(worker)
Fetching and Processing Jobs
alias BullMQ.{Job, Worker}

# Generate a unique token for this fetch operation
token = UUID.uuid4()

# Fetch the next job
case Worker.get_next_job(worker, token) do
  {:ok, nil} ->
    # No job available
    :ok

  {:ok, job} ->
    # Process the job
    case process_job(job) do
      {:ok, result} ->
        # Mark job as completed
        Job.move_to_completed(job, result, token)

      {:error, reason} ->
        # Mark job as failed
        Job.move_to_failed(job, reason, token)
    end
end
API Reference
Worker Functions
Worker.get_next_job/3
Fetches the next available job from the queue.
@spec get_next_job(worker, token, opts) :: {:ok, Job.t() | nil} | {:error, term()}
Parameters:
	worker - The worker process
	token - A unique string representing ownership of the job lock
	opts - Options:	:block - If true (default), uses BZPOPMIN to efficiently wait for a job.
If false, returns immediately with nil if no job available.
	:timeout - Timeout in seconds for blocking wait (default: 5). Only used when
block: true. After timeout, returns {:ok, nil}.



Returns:
	{:ok, job} - A job was fetched successfully
	{:ok, nil} - No job available (timeout or block: false) or worker is paused/closing
	{:error, reason} - An error occurred

Blocking Behavior:
When block: true (the default), this function uses Redis's BZPOPMIN command
to efficiently wait for jobs without polling. This is the same mechanism used
by Node.js BullMQ:
	First tries to fetch a job immediately
	If no job is available, waits using BZPOPMIN on the marker key
	When a job becomes available (marker is set), fetches and returns it
	If timeout is reached, returns {:ok, nil}

This approach is more efficient than polling because:
	No CPU cycles wasted on empty polls
	Immediate response when a job arrives
	Minimal Redis traffic

Worker.start_stalled_check_timer/1
Starts the stalled job checker. This is important for detecting jobs whose locks have expired.
:ok = Worker.start_stalled_check_timer(worker)
Worker.stop_stalled_check_timer/1
Stops the stalled job checker.
:ok = Worker.stop_stalled_check_timer(worker)
Job Functions
Job.move_to_completed/4
Moves a job to the completed state.
@spec move_to_completed(job, return_value, token, opts) :: {:ok, nil | {list(), String.t()}} | {:error, term()}
Parameters:
	job - The job struct
	return_value - The result to store with the completed job
	token - The lock token (same as used in get_next_job)
	opts - Options:	:fetch_next - If true (default), returns the next job data
	:remove_on_complete - Job removal settings



Returns:
	{:ok, nil} - Job completed, no next job
	{:ok, {job_data, job_id}} - Job completed, next job data returned
	{:error, reason} - Failed to move job

Job.move_to_failed/4
Moves a job to the failed state.
@spec move_to_failed(job, error, token, opts) :: {:ok, nil | {list(), String.t()}} | {:error, term()}
Parameters:
	job - The job struct
	error - The error (can be an Exception or a string)
	token - The lock token
	opts - Options:	:fetch_next - If true, returns the next job data (default: false)
	:remove_on_fail - Job removal settings



Job.move_to_wait/2
Moves a job back to the waiting state. Useful for rate limiting.
@spec move_to_wait(job, token) :: {:ok, non_neg_integer()} | {:error, term()}
Parameters:
	job - The job struct
	token - The lock token

Returns:
	{:ok, pttl} - Job moved back, returns rate limit TTL (or 0)

Job.extend_lock/3
Extends the lock on a job. Use this when processing takes longer than the lock duration.
@spec extend_lock(job, token, duration) :: {:ok, term()} | {:error, term()}
Parameters:
	job - The job struct
	token - The lock token
	duration - Duration in milliseconds to extend the lock

Patterns
Processing Loop
A typical processing loop that handles multiple jobs:
defmodule MyApp.ManualProcessor do
  alias BullMQ.{Job, Worker}

  def start(worker) do
    # Start stalled job checker
    Worker.start_stalled_check_timer(worker)

    # Start processing loop
    loop(worker)
  end

  defp loop(worker) do
    token = generate_token()

    case Worker.get_next_job(worker, token) do
      {:ok, nil} ->
        # No job, wait a bit
        Process.sleep(100)
        loop(worker)

      {:ok, job} ->
        process_job(job, token)
        loop(worker)
    end
  end

  defp process_job(job, token) do
    case do_work(job.data) do
      {:ok, result} ->
        Job.move_to_completed(job, result, token, fetch_next: false)

      {:error, reason} ->
        Job.move_to_failed(job, reason, token)
    end
  end

  defp do_work(data) do
    # Your processing logic
    {:ok, %{processed: true}}
  end

  defp generate_token do
    Base.encode16(:crypto.strong_rand_bytes(16), case: :lower)
  end
end
Rate Limiting
Handle rate limiting by moving jobs back to wait:
defp process_job(job, token) do
  case check_rate_limit() do
    :ok ->
      case do_work(job.data) do
        {:ok, result} ->
          Job.move_to_completed(job, result, token)
        {:error, reason} ->
          Job.move_to_failed(job, reason, token)
      end

    {:rate_limited, _delay} ->
      # Move job back to wait
      Job.move_to_wait(job, token)
  end
end
Long-Running Jobs with Lock Extension
For jobs that take longer than the lock duration:
defp process_long_job(job, token) do
  # Start a task to extend the lock periodically
  lock_task = Task.async(fn ->
    extend_lock_loop(job, token)
  end)

  try do
    result = do_long_work(job.data)
    Job.move_to_completed(job, result, token)
  rescue
    e ->
      Job.move_to_failed(job, Exception.message(e), token)
  after
    Task.shutdown(lock_task, :brutal_kill)
  end
end

defp extend_lock_loop(job, token) do
  # Extend every 10 seconds (assuming 30s lock duration)
  Process.sleep(10_000)

  case Job.extend_lock(job, token, 30_000) do
    {:ok, _} ->
      extend_lock_loop(job, token)
    {:error, _} ->
      # Lock lost, job will be picked up by another worker
      :ok
  end
end
Chained Processing with Fetch Next
Efficiently process jobs by fetching the next job with completion:
defp process_chain(worker, nil, _token) do
  # No more jobs, fetch fresh
  token = generate_token()
  case Worker.get_next_job(worker, token, block: false) do
    {:ok, job} when not is_nil(job) ->
      process_chain(worker, job, token)
    _ ->
      :done
  end
end

defp process_chain(worker, job, token) do
  result = do_work(job.data)

  # Complete job and get next in one call
  case Job.move_to_completed(job, result, token, fetch_next: true) do
    {:ok, nil} ->
      # No more jobs
      :done

    {:ok, {job_data, job_id}} ->
      # Got next job, reconstruct and continue
      next_job = Job.from_redis(job_id, job.queue_name, list_to_map(job_data),
        prefix: job.prefix,
        token: token,
        connection: job.connection
      )
      process_chain(worker, next_job, token)
  end
end
Token Management
Tokens represent ownership of a job's lock. Best practices:
	Use unique tokens - Generate a new token for each job fetch
	Keep tokens consistent - Use the same token for get_next_job, move_to_completed/move_to_failed, and extend_lock
	Don't reuse tokens across jobs - Each job should have its own token

# Good: UUID-based tokens
defp generate_token do
  UUID.uuid4()
end

# Also good: Crypto-random tokens
defp generate_token do
  Base.encode16(:crypto.strong_rand_bytes(16), case: :lower)
end
Stalled Jobs
When processing manually, enable the stalled job checker to handle jobs whose locks have expired:
# Configure stalled job behavior
{:ok, worker} = Worker.start_link(
  queue: "my-queue",
  connection: :redis,
  processor: nil,
  autorun: false,
  stalled_interval: 30_000,    # Check every 30 seconds
  max_stalled_count: 1         # Fail after 1 stall (default)
)

# Start the checker
Worker.start_stalled_check_timer(worker)
If a job's lock expires before completion:
	It's moved back to waiting (if max_stalled_count not exceeded)
	It's moved to failed (if max_stalled_count exceeded)

Comparison with Node.js
The Elixir API closely mirrors the Node.js BullMQ manual processing API:
	Node.js	Elixir
	worker.getNextJob(token)	Worker.get_next_job(worker, token)
	job.moveToCompleted(value, token)	Job.move_to_completed(job, value, token)
	job.moveToFailed(error, token)	Job.move_to_failed(job, error, token)
	job.moveToWait(token)	Job.move_to_wait(job, token)
	job.extendLock(token, duration)	Job.extend_lock(job, token, duration)
	worker.startStalledCheckTimer()	Worker.start_stalled_check_timer(worker)

See Also
	Job Cancellation - Cooperative cancellation for long-running jobs
	Rate Limiting - Built-in rate limiting support



  

    Job Cancellation

BullMQ Elixir provides cooperative job cancellation, allowing you to gracefully stop running jobs. This is useful for implementing timeouts, user-initiated cancellations, or graceful shutdown.
Overview
Job cancellation in BullMQ Elixir is:
	Cooperative: Processors must check for cancellation; BullMQ cannot forcibly stop a running function
	Efficient: O(1) cancellation checks using Erlang's native message passing
	Scalable: Zero overhead per job - tokens are just references, no ETS or GenServer
	Non-blocking: Uses receive after 0 pattern for instant checks

How It Works
When you define a processor with arity 2 (two arguments), it receives the job and a cancellation token:
processor: fn job, cancel_token ->
  # cancel_token is a reference
  # Check your mailbox for {:cancel, ^cancel_token, reason}
  {:ok, result}
end
When Worker.cancel_job/3 is called, BullMQ sends a {:cancel, token, reason} message to the task running your processor. Your processor can check for this message using various patterns.
Usage Patterns
Pattern 1: Receive After 0 (Recommended)
Best for processors that do chunked work. The receive after 0 pattern is non-blocking and O(1):
processor: fn job, cancel_token ->
  Enum.reduce_while(job.data.items, {:ok, []}, fn item, {:ok, acc} ->
    receive do
      {:cancel, ^cancel_token, reason} ->
        # Cancellation requested
        {:halt, {:error, {:cancelled, reason}}}
    after
      0 ->
        # No cancellation, process this item
        result = process_item(item)
        {:cont, {:ok, [result | acc]}}
    end
  end)
end
Key points:
	^cancel_token uses pin operator to match your specific token
	after 0 makes the receive non-blocking
	Returns immediately if no cancellation message

Pattern 2: Wrap Blocking Operations
For operations that block (HTTP calls, database queries, etc.), wrap them in a Task:
processor: fn job, cancel_token ->
  # Start the blocking operation in a Task
  task = Task.async(fn ->
    HTTPClient.post(job.data.url, job.data.body)
  end)

  # Wait for either completion or cancellation
  receive do
    {:cancel, ^cancel_token, reason} ->
      # Cancel requested - kill the task
      Task.shutdown(task, :brutal_kill)
      {:error, {:cancelled, reason}}

    {^task, {:ok, response}} ->
      {:ok, response}

    {^task, {:error, _} = error} ->
      error
  end
end
Pattern 3: Using CancellationToken.check/1
For simpler checkpoint-style cancellation:
alias BullMQ.CancellationToken

processor: fn job, cancel_token ->
  result1 = step_one(job.data)

  case CancellationToken.check(cancel_token) do
    {:cancelled, reason} -> {:error, {:cancelled, reason}}
    :ok ->
      result2 = step_two(result1)

      case CancellationToken.check(cancel_token) do
        {:cancelled, reason} -> {:error, {:cancelled, reason}}
        :ok -> {:ok, step_three(result2)}
      end
  end
end
Or use check!/1 which raises on cancellation:
processor: fn job, cancel_token ->
  CancellationToken.check!(cancel_token)
  result1 = step_one(job.data)

  CancellationToken.check!(cancel_token)
  result2 = step_two(result1)

  CancellationToken.check!(cancel_token)
  {:ok, step_three(result2)}
end
Pattern 4: Recursive Processing
For recursive algorithms:
defmodule MyProcessor do
  def process(job, cancel_token) do
    process_items(job.data.items, cancel_token, [])
  end

  defp process_items([], _token, acc), do: {:ok, Enum.reverse(acc)}

  defp process_items([item | rest], token, acc) do
    receive do
      {:cancel, ^token, reason} ->
        {:error, {:cancelled, reason}}
    after
      0 ->
        result = process_item(item)
        process_items(rest, token, [result | acc])
    end
  end

  defp process_item(item), do: item * 2
end

# Use in worker
processor: &MyProcessor.process/2
Cancelling Jobs
Cancel a Specific Job
# Cancel by job ID
:ok = Worker.cancel_job(worker, job_id, "User requested cancellation")

# Returns {:error, :not_found} if job is not active
{:error, :not_found} = Worker.cancel_job(worker, "unknown-id", "reason")
Cancel All Active Jobs
Useful for graceful shutdown:
:ok = Worker.cancel_all_jobs(worker, "Worker shutting down")
Automatic Cancellation on Lock Loss
BullMQ automatically cancels jobs when their lock renewal fails. This can happen due to:
	Network connectivity issues with Redis
	Redis server problems or restarts
	Lock TTL expired before renewal could complete

When a lock is lost, the processor receives a cancellation with reason {:lock_lost, job_id}:
processor: fn job, cancel_token ->
  Enum.reduce_while(job.data.items, {:ok, []}, fn item, {:ok, acc} ->
    receive do
      {:cancel, ^cancel_token, {:lock_lost, _job_id}} ->
        # Lock was lost - another worker may process this job
        Logger.warning("Lock lost, stopping to avoid duplicates")
        {:halt, {:error, :lock_lost}}
      {:cancel, ^cancel_token, reason} ->
        {:halt, {:error, {:cancelled, reason}}}
    after
      0 ->
        result = process_item(item)
        {:cont, {:ok, [result | acc]}}
    end
  end)
end
This prevents duplicate processing when another worker picks up the same job after the lock expires.
Backward Compatibility
Processors with arity 1 continue to work without cancellation support:
# Old-style processor - still works
processor: fn job ->
  {:ok, process(job)}
end
Best Practices
1. Check Cancellation at Safe Points
Only check for cancellation when it's safe to stop:
processor: fn job, cancel_token ->
  # Start transaction
  {:ok, tx} = Database.begin_transaction()

  try do
    # Do work within transaction
    result = do_work(tx, job.data)

    # Check cancellation AFTER transaction work but BEFORE commit
    case CancellationToken.check(cancel_token) do
      {:cancelled, reason} ->
        Database.rollback(tx)
        {:error, {:cancelled, reason}}
      :ok ->
        Database.commit(tx)
        {:ok, result}
    end
  rescue
    e ->
      Database.rollback(tx)
      reraise e, __STACKTRACE__
  end
end
2. Clean Up Resources on Cancellation
processor: fn job, cancel_token ->
  {:ok, file} = File.open(job.data.path, [:write])

  try do
    write_with_cancellation(file, job.data.content, cancel_token)
  after
    File.close(file)
  end
end

defp write_with_cancellation(file, content, token) do
  chunks = chunk_content(content)

  Enum.reduce_while(chunks, :ok, fn chunk, :ok ->
    receive do
      {:cancel, ^token, reason} ->
        {:halt, {:error, {:cancelled, reason}}}
    after
      0 ->
        IO.write(file, chunk)
        {:cont, :ok}
    end
  end)
end
3. Use Appropriate Timeouts for Tasks
processor: fn job, cancel_token ->
  task = Task.async(fn -> external_api_call(job.data) end)

  receive do
    {:cancel, ^cancel_token, reason} ->
      Task.shutdown(task, :brutal_kill)
      {:error, {:cancelled, reason}}

    {^task, result} ->
      result
  after
    30_000 ->
      # Timeout - treat as failure
      Task.shutdown(task, :brutal_kill)
      {:error, :timeout}
  end
end
Error Handling
When a job is cancelled, you can return an error or let it fail:
# Return error tuple - job will be marked as failed
{:error, {:cancelled, reason}}

# Raise exception - same result
raise "Job cancelled: #{reason}"
The job will follow normal failure/retry logic based on its configuration.
Distributed Cancellation
When running workers across multiple Elixir nodes, you can use OTP's built-in distributed messaging to propagate cancellation requests. This is more efficient and reliable than Redis Pub/Sub for Elixir-to-Elixir communication.
Using Process Groups (:pg)
The recommended approach uses Erlang's :pg module (process groups) to track all workers and broadcast cancellations:
defmodule MyApp.WorkerRegistry do
  @moduledoc """
  Registry for distributed worker cancellation using :pg process groups.
  """

  @group :bullmq_workers

  def start_link do
    # Ensure :pg is started (usually in application.ex)
    :pg.start_link(@group)
  end

  @doc "Register a worker in the process group"
  def register(worker_pid, queue_name) do
    :pg.join(@group, {__MODULE__, queue_name}, worker_pid)
  end

  @doc "Unregister a worker"
  def unregister(worker_pid, queue_name) do
    :pg.leave(@group, {__MODULE__, queue_name}, worker_pid)
  end

  @doc "Cancel a job across all workers on all nodes"
  def cancel_job(queue_name, job_id, reason \\ nil) do
    workers = :pg.get_members(@group, {__MODULE__, queue_name})

    for worker <- workers do
      Worker.cancel_job(worker, job_id, reason)
    end

    :ok
  end

  @doc "Cancel a job on workers in a specific node"
  def cancel_job(queue_name, job_id, reason, node) do
    workers = :pg.get_local_members(@group, {__MODULE__, queue_name})
    |> Enum.filter(&(node(&1) == node))

    for worker <- workers do
      Worker.cancel_job(worker, job_id, reason)
    end

    :ok
  end
end
Usage:
# In your application.ex
def start(_type, _args) do
  children = [
    {MyApp.WorkerRegistry, []},
    # ... other children
  ]

  Supervisor.start_link(children, strategy: :one_for_one)
end

# When starting a worker
{:ok, worker} = Worker.start_link(
  queue: "my-queue",
  connection: conn,
  processor: &MyProcessor.process/2
)
MyApp.WorkerRegistry.register(worker, "my-queue")

# Cancel from anywhere in the cluster
MyApp.WorkerRegistry.cancel_job("my-queue", job_id, "User cancelled")
Using GenServer.multi_call
For simpler cases, you can use GenServer.multi_call/4 to call all workers directly:
defmodule MyApp.DistributedCancellation do
  @doc "Cancel a job across all connected nodes"
  def cancel_job(worker_name, job_id, reason \\ nil) do
    nodes = [node() | Node.list()]

    # Call all workers registered with the same name across nodes
    {replies, bad_nodes} = GenServer.multi_call(
      nodes,
      worker_name,
      {:cancel_job, job_id, reason},
      5_000
    )

    case bad_nodes do
      [] -> :ok
      _ -> {:partial, replies, bad_nodes}
    end
  end
end
This requires workers to be registered with the same name across nodes.
Node.js Interoperability
If you need to cancel jobs from Node.js (or vice versa), you can create a simple Redis Pub/Sub bridge. This keeps the Elixir side clean while enabling cross-language cancellation.
Redis Pub/Sub Bridge
defmodule MyApp.CancellationBridge do
  @moduledoc """
  Bridges Redis Pub/Sub cancellation messages to Elixir workers.

  Node.js can publish to Redis, and this bridge forwards
  cancellation requests to the appropriate Elixir workers.
  """

  use GenServer
  require Logger

  @channel "bullmq:cancel"

  def start_link(opts) do
    GenServer.start_link(__MODULE__, opts, name: __MODULE__)
  end

  @impl true
  def init(opts) do
    redis_opts = Keyword.fetch!(opts, :redis)
    worker_registry = Keyword.get(opts, :worker_registry, MyApp.WorkerRegistry)

    # Start a dedicated Redis connection for Pub/Sub
    {:ok, pubsub} = Redix.PubSub.start_link(redis_opts)
    {:ok, _ref} = Redix.PubSub.subscribe(pubsub, @channel, self())

    {:ok, %{pubsub: pubsub, worker_registry: worker_registry}}
  end

  @impl true
  def handle_info(
    {:redix_pubsub, _pubsub, _ref, :message, %{channel: @channel, payload: payload}},
    state
  ) do
    case Jason.decode(payload) do
      {:ok, %{"queue" => queue, "jobId" => job_id, "reason" => reason}} ->
        Logger.info("Received cancellation from Redis: #{queue}/#{job_id}")
        state.worker_registry.cancel_job(queue, job_id, reason)

      {:ok, %{"queue" => queue, "jobId" => job_id}} ->
        Logger.info("Received cancellation from Redis: #{queue}/#{job_id}")
        state.worker_registry.cancel_job(queue, job_id, nil)

      {:error, error} ->
        Logger.warning("Invalid cancellation payload: #{inspect(error)}")
    end

    {:noreply, state}
  end

  def handle_info({:redix_pubsub, _pubsub, _ref, :subscribed, _}, state) do
    Logger.info("CancellationBridge subscribed to #{@channel}")
    {:noreply, state}
  end

  def handle_info(msg, state) do
    Logger.debug("CancellationBridge received: #{inspect(msg)}")
    {:noreply, state}
  end

  @impl true
  def terminate(_reason, state) do
    Redix.PubSub.stop(state.pubsub)
    :ok
  end
end
Node.js Side
import { createClient } from 'redis';

const CANCEL_CHANNEL = 'bullmq:cancel';

async function cancelJob(queue: string, jobId: string, reason?: string) {
  const client = createClient();
  await client.connect();

  const message = JSON.stringify({ queue, jobId, reason });
  await client.publish(CANCEL_CHANNEL, message);

  await client.quit();
}

// Usage
await cancelJob('my-queue', 'job-123', 'User cancelled');
Complete Setup Example
# In your application.ex
def start(_type, _args) do
  redis_opts = [host: "localhost", port: 6379]

  children = [
    # Worker registry for distributed cancellation
    {MyApp.WorkerRegistry, []},

    # Redis bridge for Node.js interop (optional)
    {MyApp.CancellationBridge, redis: redis_opts, worker_registry: MyApp.WorkerRegistry},

    # Your workers
    {Worker,
      name: :my_worker,
      queue: "my-queue",
      connection: redis_opts,
      processor: &MyProcessor.process/2
    }
  ]

  Supervisor.start_link(children, strategy: :one_for_one)
end
This architecture gives you:
	Elixir-to-Elixir: Fast, reliable cancellation via OTP's :pg (no Redis overhead)
	Node.js-to-Elixir: Redis Pub/Sub bridge forwards to Elixir workers
	Elixir-to-Node.js: Publish to Redis from Elixir if needed
	Separation of concerns: The bridge is optional and isolated

Performance Considerations
	Token creation: O(1) - just make_ref()
	Cancellation check: O(1) - receive after 0 scans mailbox once
	Cancel notification: O(1) - direct send/2 to task process
	Memory: One reference per active job (8 bytes on 64-bit)

The implementation has zero overhead for jobs that don't use cancellation - processors with arity 1 don't create or check tokens.


  

    Deduplication

Deduplication in BullMQ prevents duplicate jobs from being added to the queue. When a job with a deduplication ID is added, any subsequent attempts to add a job with the same ID are ignored until the deduplication condition is cleared.
Modes
BullMQ supports three deduplication modes:
Simple Mode
In Simple Mode, deduplication lasts until the job completes or fails. This is useful for long-running jobs that should not be duplicated while in progress.
# Add a job that will be deduplicated until completion or failure
BullMQ.Queue.add("tasks", "process-file", %{file: "report.csv"},
  connection: :redis,
  deduplication: %{id: "file-report.csv"}
)

# This will be ignored while the first job is still processing
BullMQ.Queue.add("tasks", "process-file", %{file: "report.csv"},
  connection: :redis,
  deduplication: %{id: "file-report.csv"}
)
Throttle Mode
In Throttle Mode, deduplication lasts for a specified TTL (time-to-live). This is useful for preventing rapid duplicate requests.
# Add a job that will be deduplicated for 5 seconds
BullMQ.Queue.add("notifications", "send-email", %{to: "user@example.com"},
  connection: :redis,
  deduplication: %{id: "email-user@example.com", ttl: 5_000}
)

# Ignored if added within 5 seconds
BullMQ.Queue.add("notifications", "send-email", %{to: "user@example.com"},
  connection: :redis,
  deduplication: %{id: "email-user@example.com", ttl: 5_000}
)

# After 5 seconds, a new job can be added
Debounce Mode
In Debounce Mode, each new job with the same deduplication ID extends the TTL and optionally replaces the job data. This is useful when you want to keep only the most recent version of a job.
# Add a job with debounce behavior
BullMQ.Queue.add("search", "update-index", %{query: "first"},
  connection: :redis,
  delay: 5_000,
  deduplication: %{
    id: "search-index",
    ttl: 5_000,
    extend: true,
    replace: true
  }
)

# This replaces the previous job and resets the TTL
BullMQ.Queue.add("search", "update-index", %{query: "updated"},
  connection: :redis,
  delay: 5_000,
  deduplication: %{
    id: "search-index",
    ttl: 5_000,
    extend: true,
    replace: true
  }
)

# Only one job will be processed, with data: %{query: "updated"}
Managing Deduplication
Get Deduplication Job ID
Find which job started the deduplication:
{:ok, job_id} = BullMQ.Queue.get_deduplication_job_id("my-queue", "dedup-id",
  connection: :redis
)

case job_id do
  nil -> IO.puts("No active deduplication")
  id -> IO.puts("Deduplication started by job: #{id}")
end
Remove Deduplication Key
Stop deduplication early, allowing new jobs to be added:
# Remove deduplication before TTL expires or job completes
{:ok, 1} = BullMQ.Queue.remove_deduplication_key("my-queue", "dedup-id",
  connection: :redis
)

# Now a new job with the same ID can be added
BullMQ.Queue.add("my-queue", "job", %{},
  connection: :redis,
  deduplication: %{id: "dedup-id"}
)
Removing Deduplication When Job Becomes Active
A common pattern is to stop deduplication as soon as a job starts processing, allowing a new job to be queued while the current one runs:
defmodule MyWorker do
  def start_link(opts) do
    processor = fn job ->
      # Stop deduplication when job starts
      if job.opts[:deduplication] do
        dedup_id = job.opts[:deduplication][:id]
        BullMQ.Queue.remove_deduplication_key("my-queue", dedup_id,
          connection: Keyword.fetch!(opts, :connection)
        )
      end

      # Process the job
      process(job.data)
      :ok
    end

    BullMQ.Worker.start_link(
      Keyword.merge(opts, [
        queue: "my-queue",
        processor: processor
      ])
    )
  end
end
Deduplication with Job Schedulers
Job schedulers don't directly support deduplication options, but you can achieve similar behavior by having the scheduler trigger a job that adds the deduplicated job:
defmodule SchedulerWorker do
  def start_link(opts) do
    processor = fn job ->
      case job.name do
        "scheduler-trigger" ->
          # Add a deduplicated job
          BullMQ.Queue.add("tasks", "actual-task", %{},
            connection: Keyword.fetch!(opts, :connection),
            deduplication: %{id: "scheduled-task", ttl: 90_000}
          )
          :ok

        "actual-task" ->
          # Process the actual task
          do_work()
          :ok
      end
    end

    BullMQ.Worker.start_link(
      Keyword.merge(opts, [
        queue: "tasks",
        processor: processor
      ])
    )
  end
end

# Set up the scheduler
BullMQ.Queue.upsert_job_scheduler("tasks", "every-minute",
  connection: :redis,
  pattern: "* * * * *",
  template: %{name: "scheduler-trigger", data: %{}}
)
Deduplication Options Reference
	Option	Type	Required	Description
	id	string	Yes	Unique identifier for deduplication
	ttl	integer	No	Time-to-live in milliseconds
	extend	boolean	No	Extend TTL on each duplicate
	replace	boolean	No	Replace job data while delayed

Best Practices
	Choose meaningful IDs: Use IDs that represent the logical operation being deduplicated, not just random values.
# Good: ID represents the operation
deduplication: %{id: "sync-user-#{user_id}"}

# Bad: Generic ID
deduplication: %{id: "job-123"}

	Use Simple Mode for critical operations: When a job absolutely must not run twice simultaneously.

	Use Throttle Mode for rate limiting: When you want to limit how often a job can be triggered.

	Use Debounce Mode for frequent updates: When multiple rapid updates should be collapsed into one.

	Consider removing deduplication on active: If you want to allow queuing the next job while the current one runs.


See Also
	Job Options - All job configuration options
	Job Schedulers - Creating recurring jobs
	Queue Events - Listen for deduplicated events



  

    Queue Events

BullMQ provides real-time event subscriptions through Redis Streams, allowing you to monitor job lifecycle events across your queue.
Overview
There are two ways to receive job events in BullMQ for Elixir:
	Worker Callbacks - Direct callbacks on the worker for jobs it processes
	QueueEvents - Centralized event listener for all jobs in a queue

Use Worker Callbacks when you want to react to events for jobs processed by a specific worker. Use QueueEvents when you need to monitor all events across a queue, regardless of which worker processes them.
Worker Callbacks (Recommended)
The simplest way to handle events is through worker callbacks:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "emails",
  connection: :my_redis,
  processor: &process/1,

  on_completed: fn job, result ->
    Logger.info("Job #{job.id} completed: #{inspect(result)}")
  end,

  on_failed: fn job, reason ->
    Logger.error("Job #{job.id} failed: #{reason}")
  end,

  on_active: fn job ->
    Logger.debug("Job #{job.id} started")
  end
)
See Workers for more details on worker callbacks.
QueueEvents
For monitoring all queue events (including jobs processed by other workers or Node.js workers), use BullMQ.QueueEvents:
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "emails",
  connection: :my_redis
)

# Subscribe the current process
BullMQ.QueueEvents.subscribe(events)

# Receive events
receive do
  {:bullmq_event, :completed, data} ->
    IO.puts("Job #{data["jobId"]} completed!")

  {:bullmq_event, :failed, data} ->
    IO.puts("Job #{data["jobId"]} failed: #{data["failedReason"]}")

  {:bullmq_event, :waiting, data} ->
    IO.puts("Job #{data["jobId"]} waiting")
end
Event Types
The following events are emitted:
	Event	Description	Data Fields
	:added	Job was added to the queue	jobId, name
	:waiting	Job is waiting to be processed	jobId
	:active	Job started processing	jobId, prev
	:progress	Job progress was updated	jobId, data
	:completed	Job completed successfully	jobId, returnvalue, prev
	:failed	Job failed	jobId, failedReason, prev
	:delayed	Job was delayed	jobId, delay
	:stalled	Job was detected as stalled	jobId
	:removed	Job was removed	jobId, prev
	:drained	Queue has no more waiting jobs	(no data)
	:paused	Queue was paused	(no data)
	:resumed	Queue was resumed	(no data)

Message Format
Events are sent as tuples with the format:
{:bullmq_event, event_type, event_data}
Where:
	event_type is an atom (:completed, :failed, etc.)
	event_data is a map with string keys

# Example completed event
{:bullmq_event, :completed, %{
  "event" => "completed",
  "jobId" => "abc123",
  "returnvalue" => "null",
  "prev" => "active"
}}
QueueEvents Options
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "my_queue",           # Queue name (required)
  connection: :my_redis,       # Redis connection (required)
  prefix: "bull",              # Queue prefix (default: "bull")
  autorun: true,               # Start listening immediately (default: true)
  last_event_id: "$",          # Start from event ID (default: "$" = new events)
  handler: MyEventHandler,     # Handler module (optional)
  handler_state: %{}           # Initial handler state (optional)
)
Multiple Subscribers
Multiple processes can subscribe to the same QueueEvents instance:
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "tasks",
  connection: :my_redis
)

# Subscribe multiple processes
BullMQ.QueueEvents.subscribe(events)  # subscribes self()
BullMQ.QueueEvents.subscribe(events, other_pid)

# Unsubscribe when done
BullMQ.QueueEvents.unsubscribe(events)
BullMQ.QueueEvents.unsubscribe(events, other_pid)
Handler Module Pattern
For more structured event handling, implement a handler module:
defmodule MyApp.QueueHandler do
  @behaviour BullMQ.QueueEvents.Handler

  require Logger

  @impl true
  def handle_event(:completed, %{"jobId" => id, "returnvalue" => value}, state) do
    Logger.info("Job #{id} completed with: #{value}")
    {:ok, state}
  end

  @impl true
  def handle_event(:failed, %{"jobId" => id, "failedReason" => reason}, state) do
    Logger.error("Job #{id} failed: #{reason}")
    MyApp.Alerts.notify_failure(id, reason)
    {:ok, state}
  end

  @impl true
  def handle_event(:drained, _data, state) do
    Logger.info("Queue drained - no more waiting jobs")
    {:ok, state}
  end

  @impl true
  def handle_event(_event, _data, state) do
    # Ignore other events
    {:ok, state}
  end
end

# Use the handler
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "tasks",
  connection: :my_redis,
  handler: MyApp.QueueHandler,
  handler_state: %{notifications_sent: 0}
)
Delayed Start
You can start QueueEvents without immediately listening for events:
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "tasks",
  connection: :my_redis,
  autorun: false  # Don't start listening yet
)

# Later, start listening
BullMQ.QueueEvents.run(events)
Closing QueueEvents
# Close the event listener
BullMQ.QueueEvents.close(events)
Example: Monitoring Dashboard
Here's an example of using QueueEvents for a simple monitoring dashboard:
defmodule MyApp.QueueMonitor do
  use GenServer

  def start_link(queue_name) do
    GenServer.start_link(__MODULE__, queue_name, name: __MODULE__)
  end

  def init(queue_name) do
    {:ok, events} = BullMQ.QueueEvents.start_link(
      queue: queue_name,
      connection: :my_redis
    )
    BullMQ.QueueEvents.subscribe(events)

    {:ok, %{
      events: events,
      completed: 0,
      failed: 0,
      active: 0
    }}
  end

  def handle_info({:bullmq_event, :completed, _data}, state) do
    {:noreply, %{state | completed: state.completed + 1, active: state.active - 1}}
  end

  def handle_info({:bullmq_event, :failed, _data}, state) do
    {:noreply, %{state | failed: state.failed + 1, active: state.active - 1}}
  end

  def handle_info({:bullmq_event, :active, _data}, state) do
    {:noreply, %{state | active: state.active + 1}}
  end

  def handle_info({:bullmq_event, _event, _data}, state) do
    {:noreply, state}
  end

  def get_stats do
    GenServer.call(__MODULE__, :get_stats)
  end

  def handle_call(:get_stats, _from, state) do
    {:reply, %{
      completed: state.completed,
      failed: state.failed,
      active: state.active
    }, state}
  end
end
Supervision
Add QueueEvents to your supervision tree:
children = [
  {Redix, name: :my_redis, host: "localhost"},

  {BullMQ.QueueEvents,
    queue: "important-queue",
    connection: :my_redis,
    handler: MyApp.ImportantQueueHandler
  }
]
Node.js Compatibility
QueueEvents is fully compatible with Node.js BullMQ. Events emitted by Node.js workers are received by Elixir QueueEvents listeners, and vice versa.
Next Steps
	Learn about Workers and their callbacks
	Set up Telemetry for metrics
	Create recurring jobs with Job Schedulers



  

    Job Schedulers

Job Schedulers allow you to create recurring jobs that run on a schedule, using either cron expressions or fixed intervals.
Overview
The BullMQ.JobScheduler module provides functions for creating, listing, and managing scheduled jobs. Unlike one-time jobs, schedulers automatically create new jobs at specified intervals.
Creating a Scheduler
Interval-based Scheduler
Create a scheduler that runs at fixed intervals:
# Run every minute
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "heartbeat",
  %{every: 60_000},  # 60 seconds in milliseconds
  "ping",            # Job name
  %{type: "health"}, # Job data
  prefix: "bull"
)

# Run every 5 seconds
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "fast-check",
  %{every: 5_000},
  "check",
  %{},
  prefix: "bull"
)
Cron-based Scheduler
Create a scheduler using cron expressions:
# Run every hour at minute 0
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "reports", "hourly-report",
  %{pattern: "0 * * * *"},
  "generate-report",
  %{type: "hourly"},
  prefix: "bull"
)

# Run every day at 9 AM
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "emails", "daily-digest",
  %{pattern: "0 9 * * *"},
  "send-digest",
  %{},
  prefix: "bull"
)

# Run every weekday at 6 PM
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "notifications", "workday-reminder",
  %{pattern: "0 18 * * 1-5"},
  "send-reminder",
  %{},
  prefix: "bull"
)
Cron Expression Format
Cron expressions follow the standard format:
┌───────────── minute (0 - 59)
│ ┌───────────── hour (0 - 23)
│ │ ┌───────────── day of month (1 - 31)
│ │ │ ┌───────────── month (1 - 12)
│ │ │ │ ┌───────────── day of week (1 - 7) (Monday to Sunday)
│ │ │ │ │
* * * * *
⚠️ Important: Weekday Compatibility with Node.js
The Elixir implementation uses the crontab library which has different weekday numbering
than Node.js's cron-parser:
	Day	Elixir (crontab)	Node.js (cron-parser)
	Monday	1	1
	Tuesday	2	2
	Wednesday	3	3
	Thursday	4	4
	Friday	5	5
	Saturday	6	6
	Sunday	7	0 (or 7)

If you're sharing schedulers between Node.js and Elixir, be aware that expressions
with weekday specifications may behave differently. Consider using interval-based
schedulers (every) for cross-platform compatibility, or adjust your cron patterns accordingly.

Examples
	Pattern	Description
	* * * * *	Every minute
	*/5 * * * *	Every 5 minutes
	0 * * * *	Every hour
	0 0 * * *	Every day at midnight
	0 9 * * 1-5	Every weekday at 9 AM
	0 0 1 * *	First day of every month
	0 0 * * 7	Every Sunday at midnight (Elixir)
	0 0 * * 0	Every Sunday at midnight (Node.js - not compatible)

Scheduler Options
{:ok, job} = BullMQ.JobScheduler.upsert(
  :my_redis,           # Redis connection
  "queue_name",        # Queue name
  "scheduler_key",     # Unique scheduler identifier
  %{
    # Required: one of pattern or every
    pattern: "0 * * * *",    # Cron pattern (mutually exclusive with every)
    every: 60_000,           # Interval in ms (mutually exclusive with pattern)

    # Optional scheduling options
    limit: 10,               # Max number of iterations
    start_date: timestamp,   # When to start (milliseconds)
    end_date: timestamp,     # When to stop (milliseconds)
    tz: "America/New_York",  # Timezone for cron patterns
    immediately: true,       # Run first job immediately (pattern only)
    offset: 5000,            # Offset for interval-based schedulers
  },
  "job_name",          # Name for created jobs
  %{data: "value"},    # Job data template

  # Job options
  prefix: "bull",
  priority: 1,
  attempts: 3,
  backoff: %{type: "exponential", delay: 1000}
)
Managing Schedulers
List All Schedulers
{:ok, schedulers} = BullMQ.JobScheduler.list(:my_redis, "my_queue", prefix: "bull")

Enum.each(schedulers, fn scheduler ->
  IO.puts("#{scheduler.key}: next run at #{scheduler.next}")
end)
Get a Specific Scheduler
{:ok, scheduler} = BullMQ.JobScheduler.get(:my_redis, "my_queue", "hourly-report",
  prefix: "bull")

if scheduler do
  IO.inspect(scheduler)
  # %{
  #   key: "hourly-report",
  #   name: "generate-report",
  #   pattern: "0 * * * *",
  #   next: 1700000000000,
  #   iteration_count: 42,
  #   template: %{data: %{type: "hourly"}, opts: %{}}
  # }
end
Count Schedulers
{:ok, count} = BullMQ.JobScheduler.count(:my_redis, "my_queue", prefix: "bull")
IO.puts("Total schedulers: #{count}")
Remove a Scheduler
{:ok, removed} = BullMQ.JobScheduler.remove(:my_redis, "my_queue", "hourly-report",
  prefix: "bull")

if removed do
  IO.puts("Scheduler removed")
else
  IO.puts("Scheduler not found")
end
Updating a Scheduler
The upsert function updates an existing scheduler if the key already exists:
# Create initial scheduler
{:ok, _} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "my-scheduler",
  %{every: 60_000},
  "job-name",
  %{version: 1},
  prefix: "bull"
)

# Update with new interval and data
{:ok, _} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "my-scheduler",
  %{every: 30_000},  # Changed interval
  "job-name",
  %{version: 2},     # Updated data
  prefix: "bull"
)
Limits and Boundaries
Iteration Limit
Stop after a certain number of executions:
# Run only 5 times
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "limited",
  %{every: 60_000, limit: 5},
  "limited-job",
  %{},
  prefix: "bull"
)
Start Date
Begin scheduling from a future date:
# Start in 1 hour
start = System.system_time(:millisecond) + 3_600_000

{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "delayed-start",
  %{every: 60_000, start_date: start},
  "job",
  %{},
  prefix: "bull"
)
End Date
Stop scheduling after a date:
# Stop after 24 hours
end_time = System.system_time(:millisecond) + 86_400_000

{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "time-limited",
  %{every: 60_000, end_date: end_time},
  "job",
  %{},
  prefix: "bull"
)
Immediate Execution
Run the first job immediately (cron patterns only):
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "immediate",
  %{pattern: "0 * * * *", immediately: true},
  "job",
  %{},
  prefix: "bull"
)
# First job runs now, then every hour
Timezone Support
Specify a timezone for cron patterns:
# Run at 9 AM New York time
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "tz-aware",
  %{pattern: "0 9 * * *", tz: "America/New_York"},
  "morning-job",
  %{},
  prefix: "bull"
)
Job Options for Scheduled Jobs
Apply job options to all jobs created by a scheduler:
{:ok, job} = BullMQ.JobScheduler.upsert(:my_redis, "my_queue", "with-options",
  %{every: 60_000},
  "job-name",
  %{data: "value"},
  prefix: "bull",
  priority: 1,           # All created jobs have priority 1
  attempts: 5,           # All jobs retry up to 5 times
  backoff: %{type: "exponential", delay: 1000},
  remove_on_complete: true
)
Scheduler Data Structure
When you retrieve a scheduler, you get:
%{
  key: "scheduler-key",           # Unique identifier
  name: "job-name",               # Job name for created jobs
  pattern: "0 * * * *",           # Cron pattern (if cron-based)
  every: 60_000,                  # Interval (if interval-based)
  next: 1700000000000,            # Next scheduled run (ms timestamp)
  iteration_count: 42,            # How many times it has run
  limit: nil,                     # Max iterations (nil = unlimited)
  start_date: nil,                # Start date constraint
  end_date: nil,                  # End date constraint
  tz: "UTC",                      # Timezone
  offset: 0,                      # Offset for interval-based
  template: %{
    data: %{...},                 # Job data template
    opts: %{...}                  # Job options template
  }
}
Processing Scheduled Jobs
Scheduled jobs are processed by workers like any other job. The worker must be running on the same queue:
# Create scheduler
{:ok, _} = BullMQ.JobScheduler.upsert(:my_redis, "maintenance", "cleanup",
  %{pattern: "0 * * * *"},
  "run-cleanup",
  %{type: "hourly"},
  prefix: "bull"
)

# Worker to process the scheduled jobs
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "maintenance",
  connection: :my_redis,
  processor: fn job ->
    case job.name do
      "run-cleanup" ->
        MyApp.Cleanup.run(job.data)
        {:ok, %{cleaned: true}}
      _ ->
        {:error, "Unknown job"}
    end
  end
)
Node.js Interoperability
There are two compatibility differences between Elixir and Node.js cron parsing:
1. Seconds Field
	Format	Elixir	Node.js
	5-field (standard)	✅ "0 9 * * *"	✅ "0 9 * * *"
	6-field (with seconds)	❌ Not supported	✅ "0 0 9 * * *"

Node.js cron-parser supports an optional seconds field at the beginning:
second minute hour day month weekday
   0      0     9   *   *     *
Elixir's crontab uses standard 5-field format only. 6-field expressions
created in Node.js will fail to parse in Elixir.
2. Sunday Numbering
	Day	Elixir	Node.js	Cross-Platform?
	Monday - Saturday	1-6	1-6	✅ Compatible
	Sunday	7	0 or 7	⚠️ Use 7

Compatible Expressions
# These 5-field expressions work identically in both:
"* * * * *"      # Every minute
"*/5 * * * *"    # Every 5 minutes
"0 * * * *"      # Every hour
"0 9 * * *"      # Every day at 9 AM
"0 0 1 * *"      # First day of month
"0 9 * * 1-5"    # Monday-Friday at 9 AM ✅
"0 9 * * 6"      # Saturday at 9 AM ✅
"0 9 * * 7"      # Sunday at 9 AM ✅ (works in both!)
"0 0 * * 6,7"    # Weekend ✅
%{every: 60_000} # Interval-based (always compatible)
Incompatible Expressions (avoid)
# These will NOT work in Elixir:
"0 0 9 * * *"    # ❌ 6-field with seconds (Node.js only)
"30 0 9 * * *"   # ❌ 6-field with seconds (Node.js only)
"0 9 * * 0"      # ❌ Sunday=0 (Node.js only, fails in Elixir)
Recommendations
	Use 5-field cron expressions (no seconds) for cross-platform compatibility
	Use 7 for Sunday instead of 0
	Use interval-based schedulers (every) when sub-minute precision is needed:%{every: 30_000}  # Every 30 seconds - works everywhere


Common Patterns
Periodic Health Checks
BullMQ.JobScheduler.upsert(:my_redis, "health", "api-health",
  %{every: 30_000},  # Every 30 seconds
  "health-check",
  %{endpoints: ["api", "db", "cache"]},
  prefix: "bull"
)
Daily Reports
BullMQ.JobScheduler.upsert(:my_redis, "reports", "daily-summary",
  %{pattern: "0 6 * * *", tz: "America/New_York"},
  "generate-daily-report",
  %{report_type: "summary"},
  prefix: "bull"
)
Cache Refresh
BullMQ.JobScheduler.upsert(:my_redis, "cache", "refresh-cache",
  %{every: 300_000},  # Every 5 minutes
  "refresh-cache",
  %{cache_keys: ["users", "products"]},
  prefix: "bull",
  priority: 10  # Low priority
)
Next Steps
	Learn about Workers to process scheduled jobs
	Set up Queue Events to monitor job execution
	Configure Rate Limiting for scheduled jobs



  

    Rate Limiting

BullMQ provides built-in rate limiting to control job processing rates.
Worker-Level Rate Limiting
Limit how many jobs are processed across all workers for a queue:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "api-calls",
  connection: :my_redis,
  processor: &MyApp.ApiCaller.process/1,
  concurrency: 10,
  limiter: %{
    max: 100,        # Maximum 100 jobs
    duration: 60_000 # Per minute (60 seconds)
  }
)
When the limit is reached, workers pause until the time window resets.
How Rate Limiting Works
	Each processed job increments a Redis counter
	The counter has a TTL equal to duration
	When counter reaches max, workers wait for TTL expiration
	Counter resets automatically when TTL expires

Time 0:00 - Counter: 0/100 - Processing
Time 0:30 - Counter: 50/100 - Processing
Time 0:45 - Counter: 100/100 - Rate limited!
Time 1:00 - Counter expires - Processing resumes
Distributed Rate Limiting
Since rate limits are stored in Redis, they work across multiple nodes:
# Node A
{:ok, worker_a} = BullMQ.Worker.start_link(
  queue: "api-calls",
  connection: :my_redis,
  processor: &process/1,
  limiter: %{max: 100, duration: 60_000}
)

# Node B (different machine, same Redis)
{:ok, worker_b} = BullMQ.Worker.start_link(
  queue: "api-calls",
  connection: :my_redis,
  processor: &process/1,
  limiter: %{max: 100, duration: 60_000}
)

# Both workers share the same rate limit counter in Redis
# Combined throughput is limited to 100/minute
Rate Limiting with Concurrency
Rate limiting works independently of concurrency:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "api-calls",
  connection: :my_redis,
  processor: &MyApp.ApiCaller.process/1,
  concurrency: 50,      # Can run 50 jobs at once
  limiter: %{
    max: 100,           # But only 100 per minute total
    duration: 60_000
  }
)
Even with 50 concurrent slots, only 100 jobs will complete per minute across all workers.
Sliding Window vs Fixed Window
BullMQ uses a fixed window rate limiter:
Fixed Window:
|-------- Window 1 --------|-------- Window 2 --------|
     [100 requests]             [100 requests]

If 100 requests happen at the end of Window 1 and 100 at the
start of Window 2, you could see 200 requests in 1 minute.
For most use cases, this is sufficient. For stricter rate limiting, consider using shorter durations:
# Stricter: 10 per 6 seconds instead of 100 per minute
limiter: %{max: 10, duration: 6_000}
Multiple Queues with Different Limits
Each queue has its own rate limit:
# API calls - 100 per minute
{:ok, api_worker} = BullMQ.Worker.start_link(
  queue: "api-calls",
  connection: :my_redis,
  processor: &process_api/1,
  limiter: %{max: 100, duration: 60_000}
)

# Email sending - 50 per minute
{:ok, email_worker} = BullMQ.Worker.start_link(
  queue: "emails",
  connection: :my_redis,
  processor: &process_email/1,
  limiter: %{max: 50, duration: 60_000}
)

# Internal processing - no limit
{:ok, internal_worker} = BullMQ.Worker.start_link(
  queue: "internal",
  connection: :my_redis,
  processor: &process_internal/1
  # No limiter option = unlimited
)
Example: External API Integration
defmodule MyApp.ApiWorker do
  def process(%BullMQ.Job{data: data}) do
    case MyApp.ExternalApi.call(data["endpoint"], data["params"]) do
      {:ok, response} ->
        {:ok, response}

      {:error, :rate_limited} ->
        # The external API rate limited us
        # Job will be retried after backoff
        {:error, "External API rate limited"}

      {:error, reason} ->
        {:error, reason}
    end
  end
end

# Configure worker to stay within external API limits
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "external-api",
  connection: :my_redis,
  processor: &MyApp.ApiWorker.process/1,
  concurrency: 5,
  limiter: %{
    max: 60,          # External API allows 60/minute
    duration: 60_000
  }
)
Monitoring Rate Limits
Use worker callbacks to monitor when rate limiting occurs:
{:ok, worker} = BullMQ.Worker.start_link(
  queue: "rate-limited-queue",
  connection: :my_redis,
  processor: &process/1,
  limiter: %{max: 100, duration: 60_000},
  on_error: fn error ->
    Logger.warning("Worker error: #{inspect(error)}")
  end
)
Or use QueueEvents to monitor the queue:
{:ok, events} = BullMQ.QueueEvents.start_link(
  queue: "rate-limited-queue",
  connection: :my_redis
)

BullMQ.QueueEvents.subscribe(events)

# Monitor events in your process
Best Practices
	Set limits below actual capacity - Leave headroom for bursts
	Use shorter windows for stricter control - 10/6s instead of 100/60s
	Monitor your limits - Use telemetry to track when limits are hit
	Consider external limits - Match your rate limit to external API limits
	Test your limits - Ensure your system behaves correctly at capacity

Next Steps
	Learn about Workers for processing configuration
	Set up Telemetry to monitor rate limiting
	Configure Job Options for retry behavior

Best Practices
1. Match Rate Limits to External APIs
# If API allows 1000 req/hour, use slightly less
limiter: %{max: 900, duration: 3_600_000}
2. Use Shorter Windows for Burst Protection
# Instead of 1000/hour, use 17/minute
limiter: %{max: 17, duration: 60_000}
3. Combine with Backoff
# If rate limited by external API, use exponential backoff
BullMQ.Queue.add(queue, "api-call", %{},
  attempts: 5,
  backoff: %{type: :exponential, delay: 5000}
)
4. Monitor Rate Limit Events
:telemetry.attach(
  "rate-limit-monitor",
  [:bullmq, :rate_limit, :hit],
  fn _event, %{delay: delay}, %{queue: queue}, _config ->
    Logger.warning("Rate limit hit on #{queue}, pausing for #{delay}ms")
  end,
  nil
)
Next Steps
	Learn about Job Flows
	Set up Telemetry
	Explore Job Schedulers

